
Алгоритм метода Гаусса с выбором главных элементов для решения алгебраических систем уравнений.
Пусть дана система линейных алгебраических уравнений. Рассмотрим расширенную матрицу, состоящую из коэффициентов системы 

и свободных членов . Обозначим [image: image2.png]el



. 
Исходная система с матрицей

путем преобразований исключения переменных приводится к равносильной системе с треугольной матрицей

Рассмотрим эти преобразования подробно. Проанализируем, например, коэффициенты при , т.е. первый столбец матрицы. Может оказаться, что максимальный коэффициент при x1 находится, например, в пятом уравнении (т.е. это a51 ). Если в системе алгебраических уравнений поменять любые два уравнения местами, то решение не изменится. Поэтому поменяем местами первое уравнение с тем уравнением, в котором находится максимальный коэффициент при x1, а затем уже будем исключать x1.
Возможен случай, когда максимальный элемент при x1 находится в первом уравнении, тогда нет необходимости менять строки местами. В качестве примера мы рассмотрели переменную x1. Её будем исключать из второго, третьего, ... , n - го уравнений.
В общем случае i - я переменная исключается из (i+1) - го, (i+2) - го, ... ,n - го уравнений. Ниже приводится описание алгоритма:
1. Из уравнений с номерами i, i+1,..., n выбираем то, в котором коэффициент при максимален по модулю, а точнее - определяем номер k этого уравнения.
2. Если i - е уравнение разделим на коэффициент при [image: image5.png]


. С помощью полученного уравнения исключаем переменную [image: image6.png]


из уравнений с номерами i+1, i+2,..., n. Здесь i изменяется от 1 до n.
На этом заканчивается так называемый прямой ход, в результате которого получается треугольная матрица с единицами по диагонали.
Исключение [image: image7.png]


из l - го уравнения осуществляется преобразованием

где j меняется от i+1 до n+1. Эти преобразования необходимо выполнить для всех l от i+1 до n, т.е. исключить [image: image9.png]


из всех последующих уравнений.
Обратным ходом вычисляем [image: image10.png]

.
2). Уравнение для определения [image: image12.png]


имеет вид

Отсюда находим

где i меняется от n-1 до 1.

Пример 1. Решить систему линейных алгебраических уравнений метод Гаусса с выбором главных элементов
1.87*x1+5.38*x2+1.03*x3+1.17*x4=1.75
7.03*x1+8.04*x2+9.05*x3+6.08*x4=3.62
1.11*x1+2.02*x2+2.03*x3-3.04*x4=2.89
3.41*x1-4.52*x2+7.28*x3+5.18*x4=0.96
Решение. 1 способ. Описанный выше алгоритм можно реализовать на языке Паскаль в следующей программе:
program mG; 
const n=4;
a: array[1..n,1..n+1] of real=((1.87,5.38,1.03,1.17,1.75),
(7.03,8.04,9.05,6.08,3.62),
(1.11,2.02,2.03,-.04,2.89),
(3.41,-4.52,7.28,5.18,0.96));
var i, j, k, l: integer;
x: array [1..n] of real;
m, z : of real;
begin
for i:=1 to n do
begin
m:=abs(a[i,i]); k:=i;
for j:=i+1 to n do
if abs(a[j,i])>m then begin
m:=abs(a[j,i]); k:=j;
end; 
if k<>i then for j:=i to n+1 do
begin
z:=a[i,j]; a[i,j]:=a[k,j]; a[k,j]:=z;
end;
z:=a[i,i];
for j:=i to n+1 do
a[i,j]:=a[i,j]/z;
for l:=i+1 to n do 
begin
z:=a[l,i];
for j:=i+1 to n+1 do 
a[l,j]:=a[l,j]-a[i,j]*z;
end; 
end;
x[n]:=a[n,n+1];
for i:=n-1 downto 1 do
begin
x[i]:=a[i,n+1];
for k:=i+1 to n do 
x[i]:=x[i]-a[i,k]*x[k];
end; 
for i:=1 to n do 
writeln ('x', i, '=', x[i]:3:3);
end. 
В результате выполнения программы получается следующее Решение: x1=9.29, x2=-1.76, x3= -4.21, x4= -1.54. Для проверки результатов вычислений, подставляем найденные , например, в четвертое уравнение и получаем 0.96. Указание. Дополните приведенную программу так, чтобы контроль результатов выполнялся в программе.
Решение. 2 способ. Описанный выше алгоритм так же можно реализовать на языке Delphi в программе:

unit gauss;

{------------------------------------------------------------}

interface

const

 ndim = 20;

type

 d_vector = array[1..ndim] of double;

 d_matrix = array[1..ndim,1..ndim] of double;

procedure leq( a : d_matrix; var b : d_vector; n : word);

implementation

{------------------------------------------------------------}

procedure leq( a : d_matrix; var b : d_vector; n : word);

{

Решение системы линейных уравнений методом Гаусса.

На входе -- матрица системы a, число уравнений n и вектор-столбец b.

На выходе решение системы содержится в векторе b.

}

var

 i,j,k,l : word;

 s,t : double;

begin

 if (n > ndim) then n := ndim;

{ Каждое уравнение делится на наибольший элемент соответствующей строки матрицы a.}

 for i := 1 to n do

  begin

   s := abs(a[i,1]);

   for k := 2 to n do

    begin

     t := abs(a[i,k]);

     if ( t > s) then s := t;

    end;

   for k := 1 to n do a[i,k] := a[i,k]/s;

   b[i] := b[i]/s;

  end;

{ ПРЯМОЙ ХОД.}

 for k := 1 to n do

  begin

{ Поиск j-ого уравнения, имеющего наибольший a[j,j].}

   j := k;

   s := abs(a[k,k]);

   for i := (k+1) to n do

    begin

     t := abs(a[i,k]);

     if ( t > s) then

      begin

       s := t;

       j := i;

      end

    end;

{ найденное уравнение меняется местами с текущим -- k-ым}

   if ( j <> k) then

    begin

     for l := k to n do

      begin

       s := a[j,l];

       a[j,l] := a[k,l];

       a[k,l] := s;

      end;

     s := b[j];

     b[j] := b[k];

     b[k] := s;

    end;

{ делим k-ое уравнение на a[k,k].}

   s := a[k,k];

   a[k,k] := 1.0;

   for l := (k+1) to n do a[k,l] := a[k,l]/s;

   b[k] := b[k]/s;

{ в этом цикле -- исключение столбца ниже диагонали.}

   for i := (k+1) to n do

    begin

     s := a[i,k];

     a[i,k] := 0.0;

     for l := (k+1) to n do a[i,l] := a[i,l] - a[k,l]*s;

     b[i] := b[i] - b[k]*s

    end;

  end;

{ ОБРАТНЫЙ ХОД.}

for i := (n-1) downto 1 do

   begin

    s := b[i];

    for k := (i+1) to n do s := s - a[i,k]*b[k];

    b[i] := s

   end

end;

{------------------------------------------------------------}

end.

